
#!/bin/csh -f

echo "argv0:$argv[0] argv1:$argv[1]"
set filename=$argv[1]
#set filename="spotters.dat"
set mainwc=`wc $filename | awk '{print $1}'`
@ num = $mainwc / 8
@ mod = $mainwc % 8
echo "# lines:\t$mainwc"
echo "modulus by 8:\t$mod\t- should be 0 ( or 1 or 7 )"
echo "# of records:\t$num"

set colwc=`grep ":" $filename | wc | awk '{print $1}'`
echo "# of colons:\t$colwc"
set pipewc=`grep "|" $filename | wc | awk '{print $1}'`
@ numpip = $pipewc / 6
echo "# of pipes:\t$numpip"

set strList="id spotterName spotterAddr spotterCity spotterPhone latitude longitude"

echo " -------------------- "
foreach txtstr ($strList)
  set tsnum=`grep $txtstr $filename | wc | awk '{print $1}'`
  if ( $txtstr == "id" ) then
    set tabs="\t\t"
  else
    set tabs="\t"
  endif
  if ( $tsnum != $num ) then
    set add=" *****"
  else
    set add=""
  endif
  echo "$txtstr$tabs$tsnum$add"
end

set numempty=`grep -v "^[a-z]" $filename | wc | awk '{print $1}'`
set numcaps=`grep "^[A-Z]" $filename | wc | awk '{print $1}'`
echo "# empty lines:\t$numempty"
echo "# lines beginning with cap:\t$numcaps"
echo ""

echo "-- All above string counts should be the same - except for # of lines "
echo "and modulus.  If not, the parameter will be denoted with ****."
echo "(This is when compared to the number of records value.)"
echo "-- If modulus is not in the specified range, then the 8-line per "
echo "spotter record format has been broken."
echo "-- Sometimes, the number of empty lines may be one greater than"
echo "the others, which usually means the end of the file has 2 blank lines"
echo "instead of one."
echo "-- If there are any lines that begin with a cap letter, then it"
echo "is possible that a cut and paste operation didn't do what was expected"
echo "and may have split a line into two."

exit

